
News  
Spike2 Version 7  

We are pleased to announce the release of Spike2 version 7. A list of the new features is 
included in the Spike2 section below. We will be writing to all registered users of Spike2 
with upgrade details in due course. 

We support running version 7 under Windows NT2000, Windows XP, Windows Vista and 
Windows 7 or Intel Macintosh running Windows. We recommend that the PC has 2GB 
minimum of RAM. 

The release of version 7 means that we will not be adding significant new features to 
version 6, but we will continue to fix bugs; there will be a 6.12 release in the next few 
weeks.  

Micro1401-3 

 
We are pleased to announce the release of the Micro1401-3, our latest high-performance 
laboratory interface, which replaces the Micro1401 mk II. Below is a comparison with the 
Micro1401 mk II: 

Micro1401-3 Micro1401 mk II 
Expandable to 64 waveform input 
channels 

Expandable  to 28 waveform input 
channels 

+-5V or +-10V operation switchable 
via software 

+-5V or +-10V operation specified 
at time of order 

 16-bit waveform output 12-bit waveform output 

4MB memory 1-2MB memory 

 

Meetings and events 
3rd Annual Canadian Neuroscience Meeting 
Westin Bayshore 
Vancouver, Canada 
May 24th – 27th 2009 

Magstim TMS Summer School 
Institute of Child Health, UCL  
London, UK 
May 29th – 30th 2009 

11th Meeting of the Portuguese Society for Neuroscience 
University of Minho  
Braga, Portugal 
June 4th – 6th 2009 

#44 ~ May 2009

Back to top

http://www.can-acn2009.org/
http://www.magstim.com/support/13960.html
http://www.spn.org.pt/


 

Physiology 2009 
University College Dublin  
Dublin, Republic of Ireland 
July 7th – 10th 2009 

 

The 6th Congress of the International Society for Autonomic 
Neuroscience 
ISAN09 is a joint meeting to be held with the European 
Federation of Autonomic Societies (EFAS) in Sydney, from 
September 1st-4th, 2009. Approx. 400 international and national 
delegates are expected, having clinical and/or basic scientific 
research interests in autonomic neuroscience. 

Plenary speakers: Darwin Berg (UCSD), Hugo Critchley (University of Sussex), Max Hilz 
(University of Erlangen-Nuremberg) and Elspeth McLachlan (POWRI) 

Further details of the meeting can be found at http://www.iceaustralia.com/isan2009/ 

Proudly supported by Cambridge Electronic Design and the Foundation for High Blood 
Pressure Research, Australia. 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.11 Signal version 4.05 

Spike2 version 5.21 Signal version 3.11 

Spike2 demo Signal demo 

 

Spike2   
 Version 7 is the latest release of Spike2 and includes many new and improved 

features, with continuous development planned through the course of the version.  

 New features include: 

• Big file format allows sampling of data file sizes up to 1TB  
• Vertical marker display option extending across entire data view 
• Copy sampling configuration as text to clipboard or Log file 

 
New sampling configuration dialog 

Back to top

http://www.physiology2009.org/
http://www.iceaustralia.com/isan2009/
http://www.iceaustralia.com/isan2009/
http://www.iceaustralia.com/isan2009/
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigw3u.shtml
CED
You can click and drag the bottom right corner of the sampling configuration in Spike2 version 7 to re-size the dialog



• Improvements to the output sequencer:  
Load/swap output sequence files during sampling  
Supports the use of #include 
Sequences of up to 8191 instructions (not original micro1401 or 
1401plus) with 256 variables (not 1401plus) 
11 new instructions including an indirect JUMP for state machines 

• Script language improvements including break and continue and ability 
to resize global arrays. 

• Virtual channels now support comparison operators 
• New linear prediction script command to predict data and generate 

power spectra using the maximum entropy method 

Files created by previous versions of Spike2 are fully compatible with version 7. 
As with previous versions, future revisions of version 7 will be freely available to 
download from our website for all registered version 7 users.  

 

  

 

Q. I am trying to control Signal sampling from an external device to set the frame 
state and trigger outputs based on a visual stimulus. The External digital states 
mode in Signal seems to be what I want to use as it will set the frame state in 
response to a given pattern on the 1401 digital inputs, but this mode does not 
give any control over the pulse outputs. Is there another way to achieve this?  

A. The text sequencer can be used to give greater control of the inputs and outputs 
of the 1401. The attached sequence file, ExtCtrl.pls, is an example text 
sequence that monitors the digital inputs and triggers one of three pulse output 
protocols for a set input pattern. Each of the pulse protocols is also used to set 
the corresponding frame state. This example sequence could easily be extended 
or modified to include additional pulse outputs or respond to different bit patterns 
on the digital inputs. 

 
Using a text sequence in Signal outputs tab  

Scripts: Spike2  
Q. I am using a TextMark channel to mark episodes of EEG activity in response to 

certain stimuli. I would then like to generate power spectra of these marked 
sections only, much like using the gated analysis option found in the PSTH result 
view analysis. Is there any way to do this in Spike2? 

A. The attached script, GatedPower.s2s, allows the user to generate a ‘gated’ 
power spectrum where the sections of data included in the result are defined by a 
TextMark channel. The script also gives the option of overlapping the FFT blocks 
within each time range. 

Back to top

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

ExtCtrl.pls


'$GatedPower.s2s|power spectrum of selected time ranges
'===========================================================================================================
'	          CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB6 0FE, UK
'===========================================================================================================
'Copyright © Cambridge Electronic Design. All rights reserved.
'last modified:  30/March/2009.
'Author:	GH.

'SOFTWARE REQUIRED:
'Spike 2 version 6.06 or higher
'The script uses functions contained in the file <ghutils.s2s> located in the <include> folder
'of the Spike2 root directory.
'Save this file in the Spike2 scripts folder for correct operation.

'HEALTH WARNING:
'The script is a <work in progress> and is offered without guarantees.
'Please report any problems to:	geoff@ced.co.uk

'OVERVIEW
'This script generates a gated power spectrum of a selected channel. The episodes to be included in the power spectrum
'must be marked in a TextMark channel drawn in States mode. Some scripts such as the LevelIndicator.s2s and RangeIndicator.s2s 
'create a range marker of this type. Alternatively, you can create a channel of this type using the Spike 2 menus.
'Simply create a TextMark memory buffer and then use the <Add items> option to add Textmarks at the start and end of the desired time ranges.
'Use a non-zero marker code for the onset of a range and zero for the end.

'The script has an option to overlap the FFT blocks within each time range so that all of the data is included. The overlap is as close to 50%
'as possible given the requirement for equal representation of each time range.

'KEYWORDS
'gated power spectrum, overlap FFT blocks, state channel, 

'USER GUIDE
'The script gives you the option to add a button, <GatedPow>, to the Script Bar so that you can operate the script with one click in future.
'Comment out the line in the script between dashes below if you don t want a button and you do not want to be nagged about it.

'The script operates on the front view. If the front view is not a time view, it searches for an open time view and works on the 
'first one that it finds. If no time view is found, the script closes with a warning message.
'If a time view is found then a <Set Up> dialog opens. Here, you select the following settings:

'Waveform channel: the channel for power measurements (must be visible)
'Range channel:    a visible TextMark channel in States mode with markers for the episodes of interest
'Range code:         the code for the marker that represents the start of a time range. Any other markers will be interpreted as the end of a range.
'Start/End time       Start and end time of the range to process. You can user Csrs 1 and 2 to mark this range if you wish.
'	                         Processing starts at the onset of the first marked range after the start time.

'FFT size:         The FFT size determines the time resolution of the power plot and the number of bins. The details are shown in the <Info> box
'                     at the bottom of the dialog. Note that time ranges shorter than the FFT duration shown in this box will NOT be processed.
'Window:      -see Spike2 help under the heading <Power spectrum> for details
'Overlap blocks    Check this box to overlap fft blocks by at least 50% and ensure that all of the data in each time range is included.
'                     Overlapping FFT blocks also helps to compensate for the effects of applying a window.
'                     If you do not overlap FFT blocks then some time at the end of each marked range (max 0.999*FFT duration) will not be included
'                     in the power spectrum.

'When you are ready, click on <OK> to generate the power spectrum. The title of the spectrum will include the name of the source data file
'and the marker code used to mark the onset of ranges. Text below the x-axis of the spectrum shows the overall time range that was selected,
'the time that was actually processed and whether or not epochs were overlapped. Use the options on the Spike2 File menu to save a power spectrum.

'NOTES
'The <OK> button of the dialog will be disabled if a range marker code is set but there are no visible candidate range marker channels.
'However, if there is no range marker channel and the marker code is set to 0, you can perform a power spectrum on the entire time range
'between start and end. You could use this option to perform an un-gated power spectrum with overlapped FFT blocks, an option that is not
'available via the drop-down menus at the time of writing.

'A warning message will display if some of the episodes in the time range were too short to be included in the power spectrum.
'The message gives the number of episodes and the duration of data that was not used.

#include <ghutils.s2s>
var root$,set$;
var timevh%;

var txt%[2];
var fftlst$[11],fftndx%,wdwndx%;	                      'gated power spectrum dialog parameters
fftlst$[0]:="16"; fftlst$[1]:="32"; fftlst$[2]:="64"; fftlst$[3]:="128"; fftlst$[4]:="256"; 
fftlst$[5]:="512"; fftlst$[6]:="1024"; fftlst$[7]:="2048"; fftlst$[8]:="4096"; fftlst$[9]:="8192";  fftlst$[10]:="16384";

root$:=FilePath$(2);	                                  'path to the Spike2 root folder
set$:="GatedPow|"+root$+"\scripts\\GatedPower.s2s|Power spectrum gated by a State channel";	'label for script bar button
'--------------------------
AddScriptBarBtn%(set$);	                               'comment this line if you do not want a button on the Script Bar
'--------------------------
FindView(timevh%,0);	                                  'find an open time view and bring it to the front
if ViewKind(timevh%)<>0 then
	Message("Error|Front view is not a time view.");
	halt
endif;
View(timevh%);
GatedPower%();	                                        'power spectrum of selected time ranges of selected channels
halt

func GatedPower%()	                                  'generate power spectrum for selected vigilance state
var ok%,wdw$[10],st,et,ech%,mcode%,eegchan%,fftsz%;
var dupch1%,dupch2%,ttl$,pos%,powvh%,maxT,t1,t2;	    
var done%:=0,dur;
var overlapchk%:=1,ovlp$;
var nmissed%:=0,tmissed:=0.0;

wdw$[0]:="None"; wdw$[1]:="Hanning"; wdw$[2]:="Hamming"; wdw$[3]:="Kaiser 30dB";  wdw$[4]:="Kaiser 40dB"; 
wdw$[5]:="Kaiser 50dB"; wdw$[6]:="Kaiser 60dB"; wdw$[7]:="Kaiser 70dB"; wdw$[8]:="Kaiser 80dB"; wdw$[9]:="Kaiser 90dB";
View(timevh%);
et:=MaxTime();
DlgCreate("Power spectrum set up",50,10);
DlgAllow(511,0,Chgatedpow%);
DlgChan(1,"Waveform channel:",1+512+2048);	          'visible waveform/realwave chans
DlgChan(2,"Range channel:",32+2048);	                'visible textmark chans
DlgInteger(3,"Range code|Set zero to combine all ranges",0,10,0,0,1);
DlgXValue(4,"Start time (s):");
DlgXValue(5,"End time (s):");
DlgList(6,"FFT size:",fftlst$[],11);
DlgList(7,"Window:",wdw$[],10);
DlgGroup("Info",2,9,44,2);
txt%[0]:=DlgText("",5,9.6,40);
txt%[1]:=DlgText("",5,10.3,40);
DlgCheck(8,"Overlap FFT blocks",25.5,8);
ok%:=DlgShow(eegchan%,ech%,mcode%,st,et,fftndx%,wdwndx%,overlapchk%);
if not ok% then return 1; endif;	                      'cancelled
Yield();	                                              'time for dialog to close
View(timevh%);

var npts%;

fftsz%:=Val(fftlst$[fftndx%]);	                      'create power spectrum
maxT:=MaxTime();
ttl$:=WindowTitle$();
pos%:=InStr(ttl$,".smr");
ttl$:=Left$(ttl$,pos%-1);
powvh%:=SetPower(eegchan%,fftsz%,wdwndx%);
if overlapchk% then ovlp$:="(ovlp)"; endif;	          'label spectra with overlapping epochs
WindowTitle$(Print$("Pw rnge %d (%s)",mcode%,ttl$));
Window(0,0,40,40);
if mcode% = 0 then	                                  'process whole time range
	if overlapchk% then
		ProcessOverlapped(timevh%,eegchan%,fftsz%,st,et,powvh%);
	else            
		Process(st,et,0,1);	                            'process end to end
	endif;      
	dur:=et-st;	                                        'time processed
else	                                                 'process selected state
	View(timevh%);
	dupch1%:=ChanDuplicate(ech%);	                      'copy of epoch marker channel for onset of selected state
	dupch2%:=ChanDuplicate(ech%);	                      'copy of epoch marker channel for end of selected state
	MarkMask(dupch1%, 0);	                            'set mode 0
	MarkMask(dupch1%,-1, 1,-1);	                      'include everything (reset)
	MarkMask(dupch1%, 0, 0, -1);	                      'exclude everything in layer 0
	MarkMask(dupch1%, 0, 1, mcode%);	                   'include selected stage
	MarkMask(dupch2%, 0);	                            'set mode 0
	MarkMask(dupch2%,-1, 1,-1);	                      'include everything (reset)
	MarkMask(dupch2%, 0, 0, mcode%);	                   'exclude selected stage
	t2:=st-0.01;	                                     'make sure we include epoch at start time
	repeat
		View(timevh%);
		t1:=NextTime(dupch1%,t2);	                      'onset of selected state
		if t1 >=et then	                               'no epochs of selected type in time range
			done%:=1;	                                  'break out of the repeat loop
			View(powvh%);
			FileClose(0,-1);	                            'close empty power spectrum window
			Newsflash("No epochs of selected type in time range",0,0,0,1);        
		else	                                           'process new time range
			t2:=NextTime(dupch2%,t1);	                   'end of episode in selected state
			npts%:=(t2-t1)/BinSize(eegchan%);	          'trap ranges too short to analyse here
			if fftsz%<=npts% then	                      'this episode is long enough to process
				if t2 > et or t2 < 0 then t2:=et; done%:=1; endif;	'stop at end of time range if it occurs within the current stage
				if t1 > 0.0  and t1 < t2 then
					View(powvh%);
					if overlapchk% then
						ProcessOverlapped(timevh%,eegchan%,fftsz%,t1,t2,powvh%);
					else            
						Process(t1,t2,0,1);	                'process current time range
					endif;           
					dur+=t2-t1;	                            'total time processed into power spectrum 
				else
					done%:=1;
				endif;
			else
				nmissed%+=1;	                            'count episodes that were to short to process
				tmissed+=t2-t1;	                         'duration of unprocessed data
			endif;			
		endif;
	until done%;
	View(timevh%);
	ChanDelete(dupch1%);
	ChanDelete(dupch2%);
endif;
if ViewKind(powvh%)=4 then
	View(powvh%);
	XTitle$(Print$("Time range: %.2f to %.2f s. Time processed: %.1f s. %s",st,et,dur,ovlp$));
	XRange(0.0,35.0);
	FrontView(powvh%);
endif;
if nmissed%>0 then
	Message("Warning| %d episodes were too brief to process (% .3g s).",nmissed%,tmissed);
endif;
return 1;
end;

func ChGatedPow%(item%)
var txt$,txt2$,ul,nbins%,binw;
var en%,dur;

View(timevh%);
if item%=0 or item%=2 or item%=3 then
	if DlgValue(2)=0 and DlgValue(3)<>0 then
		en%:=0;
	else
		en%:=1;
	endif;
	DlgEnable(en%,-1);	                               'dis-able OK button if no range marker and non zero code
endif;

if item%=0 or item%=1 or item%=6 then
	ul:=0.5/BinSize(DlgValue(1));	                      'half sampling rate of selected channel
	nbins%:=Val(fftlst$[DlgValue(6)])/2.0;
	binw:=ul/nbins%;
	dur:=Val(fftlst$[DlgValue(6)])*BinSize(DlgValue(1));
	txt$:=Print$("Result will show frequencies from 0 to %.1f Hz",ul);
	txt2$:=Print$("in %d bins of width %.3f Hz. FFT dur: %.3gs",nbins%,binw,dur);
	DlgValue$(txt%[0],txt$);
	DlgValue$(txt%[1],txt2$);
endif;
return 1;
end;

func ProcessOverlapped(tvh%,chan%,fftsz%,t1,t2,powvh%);	'process epochs with at least 50% overlap
'Arguements:
'tvh%: handle of time viewhandle
'chan%: waveform channel to analyse
'fftsz%: fft block size
't1,t2: start and end of time range to Analyse
'powvh%: view handle of power spectrum window to hold the results

var fftdur,trnge,nblocks,f,i%;
var ovlp;	                                           'overlap in percent between FFT blocks
var nbl%,shift;	                                     'number of fft blocks and offset between epochs when overlapping
var st,nd;

fftdur:=fftsz%*View(tvh%).BinSize(chan%);	             'get duration of fft block
trnge:=t2-t1;	                                        'epoch duration
nblocks:=trnge/fftdur;	                               'time range expressed as nr of FFT blocks
f:=Frac(nblocks);	                                     'fraction of epoch left over
nblocks:=Trunc(nblocks);	                            'whole number of epochs in time range
docase	                                              'calculate nr of FFT blocks needed for at least 50% overlap
case f = 0 then	                                     'time range is an exact number of epochs
	nbl%:=2.0*nblocks-1;
case f <= 0.5 then	                                  'less than half of one FFT block left over
	nbl%:=2.0*nblocks;
else
	nbl%:=2.0*nblocks+1;	                               'more than half an FFT block remaining
endcase;
shift:=(trnge-fftdur)/(nbl%-1);	                      'time shift for each successive epoch
ovlp:=(fftdur-shift)*100.0/fftdur;	                   'percent overlap of each epoch
st:=t1;
nd:=t1+fftdur;
View(powvh%);
for i%:=1 to nbl% do
	Process(st,nd,0,1);
	st+=shift;
	nd+=shift;
next;
return 1;
end;



CED
GatedPower.s2s

CED
You can run output sequences continuously so that outputs can be used between sampled frames in Signal using the Free run without restarts option



Scripts: Signal  
Q. I am exporting data files to Matlab for further analysis and was hoping that there 

was a script available that would let me export batches of files, from a given 
directory for example. 

A. The attached script, BatchExportMatlab.sgs, is an example batch 
processing script that will export all data files from a selected directory to Matlab 
compatible format. The script could be modified to export to any supported file 
type in Signal. 

Did you know…? 
The CED forum includes a search option allowing the user to find posts and answers to 
specific questions and topics.  

Recent questions   
Q. I am currently writing a script to automate control of my experiment and on-line 

analysis. The script is linked to a sequencer file and triggers three different pulse 
protocols. I would like to apply these randomly but I am having some difficulty in 
randomising the order of the stimulus types in the script. Can you give me any 
examples of how to do this? 

A. The following script section is an example of using the ArrSort() function to 
randomise an array containing stimulus types. 
var stims[60];                ‘Array for stimulus types 
var randoms[60];              ‘Array for random numbers 
 
ArrConst(stims[:20],1);       ‘Fill first 20 elements with type 1 
ArrConst(stims[20:20],2);     ‘..next 20 with type 2.. 
ArrConst(stims[40:20],3);     ‘..last 20 with type 3 
 
Rand(randoms[]);           ‘Fill randoms with random numbers 
ArrSort(randoms[],0,stims[]); ‘Sort arrays 

This example uses three stimulus types (1,2 and 3) and sets them to be used 20 
times each. The first array is filled with 20 each of the stimulus types and a 
second array is filled with random numbers using the Rand() command. The 
ArrSort() function is then used to sort the arrays so that the corresponding 
elements of the randoms and stims arrays are in the same order. The end result 
is that the randoms array is sorted in ascending order while the stims array 
holding the stimuli types is now in a random order. 

User group      
Try the new CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

'$BatchExportMatlab|Script to export all files from a directory using the default matlab export settings

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 22/05/09

var files$[20];		'Declare array to hold up to 20 files
var chans%[32];		'Array to hold channels in files
var numFiles%;			'Number of files found
var c%;				'counters to loop through files and channels
var numChans%;			'Number of channels in current file
var success%;			'Test if file opened correctly
var drive$,path$,name$;     'Strings to hold destination for new file
var msg$;       'Toolbar message for current file

FilePathSet("",0,"Select directory of files for analysis");		'Prompts user to select directory
numFiles%:= FileList(files$[],0);		'Fill array with up to 20 files and return number of files found
if numFiles%<1 then							'If no files found...
	Message("No data files in directory!");		'Display message and end script
    halt;endif;

for c%:=1 to numFiles% do			'Loop through each file
	success%:=FileOpen(files$[c%-1],0);		'Check it opened correctly

	if success% > 0 then			'If it opened correctly..
        drive$:= FileName$(1);
        path$:= FileName$(2);
        name$:= FileName$(3);
        msg$:= Print$("Exporting file %d: ",c%);
        ToolbarText(msg$+name$);
        
        ChanList(chans%[],1);		'Get a list of the channels in the file
        ExportChanList(chans%[0]);
        ExportFrameList(-1);
        ExportTimeRange(0,MaxTime(),1);
        FileExportAs(drive$+path$+name$+".mat",100,0,"",0,"UseSName=1|UseCName=0|WaveOpts=0|WaveData=2|Compat=0"); 'Export with default matlab settings
        FileClose(0,-1);			'Close file to keep screen tidy
	else;
		Message("Could not open file %s",files$[c%]);	'If file not opened, display error message
	endif;
	
next;


CED
BatchExportMatlab.sgs

